               ==================================================================================

               ==================================================================================

                      DOKUMENTIERTES LISTING ZU „NULL GRAD NORD“ (Textadventure von AXIS)

                                            H.t.W  -  2006-05-29

               ==================================================================================

               ==================================================================================

0 A=A+1:ON A GOSUB 45000,45002,45003,45004:POKE 53265,PEEK(53265)AND239

1 PRINT"“":POKE 53280,0:POKE 53281,9:PRINT :GS=18:YD=13:GOSUB 42000:GOTO 5

2 PRINT TAB(1);"

<SPACE>’";

3 GETK$:IF K$<>" " THEN 3

4 GOSUB 32765:RETURN

5 GOSUB 32700:GOSUB 32600:GOTO 32400

7 REM VERSION 8.9.85

8 REM

=========================================

Gegenstände die im Inventar getragen werden können

-----------------------------------------

10 DATA 40,STANGE,46,TICKET,47,SCHAUFEL,49,GELD,50,FEUERZEUG,51,BECHER

11 DATA 52,FALLSCHIRM,53,MESSER

12 DATA 54,GURT,55,HAMMER,56,DYNAMIT,57,SEIL,58,LAMPE,59,ZELT

13 DATA 60,CAMPINGKOCHER,70,AUSWEIS,42,KARTE,71,SCHLUESSEL

=========================================

im Wohnzimmer deines Hauses

1000 D$="TXT1,S,R":PC=1:GOTO 9500

-----------------------------------------

der Flur deines Hauses

1050 D$="TXT2,S,R":GOSUB 32650:PC=2

1065 OT=0:IF ASC(MID$(G$,71,1)+CHR$(0))=0 THEN OT=1700

1070 GOSUB 32580

1080 ZL=1090:GOTO 32300

1090 GOSUB 32740:GOTO 1080

-----------------------------------------

die Küche deines Hauses
1100 D$="TXT4,S,R":PC=4:GOTO 9500

-----------------------------------------

die Toilette deines Hauses

1200 D$="TXT3,S,R":PC=3:GOTO 9500

-----------------------------------------

Badezimmer deines Hauses

1300 D$="TXT9,S,R":PC=9:GOTO 9500

-----------------------------------------

der Flur im Obergeschoss deines Hauses

(wenn W!=OEFFNE + W2=LUKE + LK(Luke)=0 + in G$ (Inventar) = an 40. Stelle=0 dann LK = 1 (offen)

(im Inventar wird immer wenn ein Gegenstand vorhanden ist die koresponierende Stelle von G$ auf 0 gesetzt)

(STANGE = NR 40)

1400 PC=5:D$="TXT5,S,R":IF LK=1 THEN D$="TXT7,S,R"

1415 GOSUB 32650

1425 GOSUB 32580

1430 ZL=1440:GOTO 32300

1440 IFW1=2ANDW2=41ANDLK=0ANDASC(MID$(G$,40,1)+CHR$(0))=0 THEN LK=1:GOTO 1400


1450 IF W1=11 AND W2=63 THEN GOSUB 1470:FL=1

1455 IFW1=2ANDW2=41 AND LK=0 AND ASC(MID$(G$,40,1)+CHR$(0))<>0 THEN GOSUB 1480

1460 GOSUB 32740:GOTO 1430

aus Fenster sehen

1470 D$="TXT6,S,R":GOSUB 32650

1475 GOSUB 32765:POKE 214,YD+6:SYS 58640:GOSUB 2:GOTO 1400

1480 GOSUB 32765:POKE 214,YD+6:SYS 58640:PRINT " DU BIST NICHT GROSS GENUG!"

1481 GOSUB 2

1485 GOSUB 32765:FL=1:RETURN

---------------------

im Schlafzimmer

1500 D$="TXT8,S,R":PC=8:GOTO 9500

--------------------------------

am Dachboden

(W1=Wort1, W2=Wort2, 

(karte zerfällt beim nehmen zu Staub

1600 D$="TXT10,S,R":GOSUB 32650:PC=10
Bild+Text10 laden
1620 GOSUB 32580
mögliche Richtungen zeigen
1630 ZL=1640:GOTO 32300
auf Eingabe warten
1640 IF W1=1 AND W2=42 AND FL=1 THEN GOSUB 40000
KARTE + NEHMEN + ?? = G$ an 42.St. Chr$200
1650 IF W1=11 AND W2=42 AND ASC(MID$(G$,42,1)+CHR$(0))<>200 THEN 1680
KARTE + ANSEHEN + noch nicht genommen dann
1660 GOSUB 32740:GOTO 1630
zu “Karte ansehen“
1680 D$="TXT28,S,R":GOSUB 32650
Bild karte ansehen – sieht alt aus
1685 GOSUB 32765:POKE 214,YD+6:GOSUB 2:GOTO 1600

--------------------------------

vor deinem Haus auf der Straße

1700 D$="TXT11,S,R":GOSUB 32650:PC=11

1715 IF ASC(MID$(G$,71,1)+CHR$(0))=0 THEN WE=1050
nur wenn Schlüüssel im Inv daan Richtung=W
1720 GOSUB 32580
(Bemerkung HA :)), guter Gag)
1730 ZL=1740:GOTO 32300

1740 GOSUB 32740:GOTO 1730

immer noch auf der Straße

1800 D$="TXT13,S,R":PC=13:GOTO 9500

====================================================
Haltestelle 2 – (auf der Straße verlaufen – ENDE)

1900 D$="TXT12,S,R":GOSUB 32650

1930 GOTO 32500

====================================================
vor dir bekannten haus

2000 D$="TXT14,S,R":GOSUB 32650:PC=14
Bild+Text laden, 
2020 GOSUB 32580
Richtungen anzeigen
2030 ZL=2040:GOTO 32300
auf Text warten
2040 IF W1=3 AND W2=64 THEN 2100
HAUS+GEHEN dann zu Bild „Haustür“

2050 GOSUB 32740:GOTO 2030

vor der Haustür des bekanntes Hauses

2100 D$="TXT15,S,R":GOSUB 32650:PC=15
Bild+Text laden,
2120 GOSUB 32580

2130 ZL=2140:GOTO 32300

2140 IF W1<>4 OR W2<>44 OR ASC(MID$(G$,49,1)+CHR$(0))=0 THEN GOTO 40030
BETAETIGEN+KLINGEL+schonGeldbekommen(49=0)
2150 D$="TXT16,S,R":GOSUB 32650:PC=16
dann Tür Verschl sonst zu Bild Feund gibt
2160 GOSUB 32765
“Freund gibt Geld” Bild laden
2165 IF GG=4 THEN 40040
Gegensrände im Inv schon 4 dann kein Geld
2170 G$=LEFT$(G$,48)+CHR$(0)+RIGHT$(G$,LEN(G$)-49):GG=GG+1
sonst Geld ins Inv + Anz Geg im Inv +1
2175 POKE 214,YD+7:GOSUB 2

2180 GOTO 2100

====================================================
Haltestelle 1 – (wieder bei der Haltestele zu Hause)

2200 D$="TXT17,S,R":GOSUB 32650:PC=17

2220 GOSUB 32580
mögliche Richtungen anzeigen
2230 ZL=2240:GOTO 32300
zu “Warten auf Eingabe”
2240 IF W1<>5 OR W2<>45 THEN GOSUB 32740:GOTO 2230

----------------------

2250 REM  Die Busfahrt

----------------------

2260 D$="TXT29,S,R":GOSUB 32650:GOSUB 32765
Bild+Text 29 laden, Eingabemaske löschen
2265 POKE 214,YD+6

2275 HT=1:WT=0:GOSUB 3:D$="TXT30,S,R":GOSUB 32650
auf Space warten, Bild+Text30 laden
2276 PRINT“....HALTESTELLENANZEIGE AUSGEFALLEN“:
!! neu eigefügt

2280 GETA$:IFA$=" " THEN 2300
!! Sprungmarke behalten
(2280 POKE 214,YD+7:SYS 58640:PRINT " HALTESTELLE ";HT)
!! diese Zeile gelöscht, da Absturz bei jeder !!
2281 WT=WT+1:IF WT=8 THEN HT=HT+1: IF HT=6 THEN HT=1
!! Änderung des Cursors (keine Ahnung, warum  !!

2282 IF WT=8 THEN GOSUB 2286
!!
2284 IF WT=8 THEN WT=0:
Haltestellen erhöhen
2285 GOTO 2280
!!

2286 POKE53380,1:POKE53280,0
!!

2287 FORI=1TO100:NEXTI
!!

2268 RETURN
!! neu eigefügt (Rahmen blitzt bei jeder Haltest. auf
---------------------

Haltestelle 5

2290 D$="TXT31,S,R":GOSUB 32650:GOTO 32500

---------------------

Space gedrückt an Haltestelle X

2300 REM

2301 REM

2302 REM


2303 REM


2305 REM


2310 GOSUB 32765:ON HT GOTO 2200,1900,2400,2500,2290
ausgestiegen bei Haltestelle 1,2,3,4,5
---------------------

Haltestelle 3

2400 D$="TXT18,S,R":GOSUB 32650:PC=18

2420 GOSUB 32580

2430 ZL=2440:GOTO 32300

2440 IF W1<>5 OR W2<>45 THEN GOSUB 32740:GOTO 2430

2450 GOTO 2250

---------------------

Haltestelle 4

2500 D$="TXT21,S,R":GOSUB 32650:PC=21

2520 GOSUB 32580

2530 ZL=2540:GOTO 32300

2540 IF W1<>5 OR W2<>45 THEN GOSUB 32740:GOTO 2530

2550 GOTO 2250

====================================================
weiter ins Elendsviertel

2600 D$="TXT19,S,R":PC=19:GOTO 9500

zu weit gegangen im Elendsviertel

2700 D$="TXT20,S,R":GOSUB 32650

2720 REM

2730 GOTO 32500

====================================================
im Flughafen verlaufen

2800 D$="TXT22,S,R":GOSUB 32650:GOTO 32500

im Flughafengelände

2900 D$="TXT23,S,R":GOSUB 32650:PC=23

2920 GOSUB 32580

2930 ZL=2940:GOTO 32300

2940 IF W1=3 AND W2=61 THEN 3000
GEHEN+SCHALTER dann zu Bild “vor dem Schalter”
2950 GOSUB 32740:GOTO 2930

vor dem Schalter im Flughafen

3000 D$="TXT24,S,R":GOSUB 32650:PC=24

3020 GOSUB 32580

3030 ZL=3040:GOTO 32300

Schalter am Flughafen

(W1=KAUFEN + W2=TICKET + im Inv 49.Stelle=0 (Geld vorhanden) dann Ticket gekauft)

3040 IF W1<>7 OR W2<>46 OR ASC(MID$(G$,49,1)+CHR$(0))<>0 THEN 40050

3050 GOSUB 32765:POKE 214,YD+6:SYS 58640

3052 PRINT " DU  HAST EIN  TICKET NACH  AMERIKA GE-"

3055 PRINT " KAUFT.":GOSUB 2

3060 G$=LEFT$(G$,48)+CHR$(63)+RIGHT$(G$,LEN(G$)-49)

3065 G$=LEFT$(G$,45)+CHR$(0)+RIGHT$(G$,LEN(G$)-46)

3070 GOTO 3030

Automant nit Aufschift Ticket

(ohne Ticket = 46 kein Durchgang)

3100 D$="TXT25,S,R":GOSUB 32650:PC=25

3120 GOSUB 32580

3130 ZL=3140:GOTO 32300

3140 IF W1<>6 OR W2<>46 OR ASC(MID$(G$,46,1)+CHR$(0))<>0 THEN 40060

3145 REM

du gehst Richtung Flugzeug

3150 D$="TXT26,S,R":GOSUB 32650:PC=26:GOSUB 32765

3155 POKE 214,YD+6:SYS 58640:GOSUB 2

Flugzeug stürzt ab

3160 D$="TXT27,S,R":GOSUB 32650:PC=27

3162 REM

nach dem Absturz wird das Inventar verändert !!

3165 G$=LEFT$(G$,45)+CHR$(1)+RIGHT$(G$,LEN(G$)-46):GG=GG-1
Ticket wird verloren
3166 IF MID$(G$,71,1)+CHR$(0)=CHR$(0) THEN GG=GG-1
Schlüssel
3167 G$=LEFT$(G$,70)+CHR$(255)+RIGHT$(G$,LEN(G$)-71)
wird verloren
3168 GOSUB 32762

3170 GOTO 3200
zu Teil 2 des Adventures

==================================================

TEIL 2 DES ADVENTURES

---------------------

Im Norden die Reste des Flugzeugs

3200 D$="TXTA21,S,R":PC=61:GOTO 9500

beim Flugzeug (dunkel)

3300 D$="TXTA8,S,R":GOSUB 32650:PC=48

3320 GOSUB 32580

3330 ZL=3340:GOTO 32300

Treibstoff explodiert
3340 IF W1<>17 OR W2<>50 OR MID$(G$,50,1)<>CHR$(0) THEN 40070

3350 REM

3355 REM

3360 D$="TXTA9,S,R":GOSUB 32650:PC=49

3365 REM

3370 GOTO 32500

die eiswüste

3400 D$="TXTA20,S,R":PC=60:GOTO 9500

3500 D$="TXTA22,S,R":PC=62:GOTO 9500

3600 D$="TXTA26,S,R":PC=66:GOTO 9500

3700 D$="TXTA16,S,R":PC=56:GOTO 9500

3800 D$="TXTA19,S,R":PC=59:GOTO 9500

3900 D$="TXTA25,S,R":PC=65:GOTO 9500

4000 D$="TXTA11,S,R":PC=51:GOTO 9500

4100 D$="TXTA10,S,R":PC=50:GOTO 9500

4200 D$="TXTA12,S,R":PC=52:GOTO 9500

4300 D$="TXTA13,S,R":PC=53:GOTO 9500

4400 D$="TXTA14,S,R":PC=54:GOTO 9500

4500 D$="TXTA15,S,R":PC=55:GOTO 9500

4600 D$="TXTA17,S,R":PC=57:GOTO 9500

4700 D$="TXTA18,S,R":PC=58:GOTO 9500

4800 D$="TXTA23,S,R":PC=63:GOTO 9500

4900 D$="TXTA24,S,R":PC=64:GOTO 9500

5000 D$="TXTA27,S,R":PC=67:GOTO 9500

5100 D$="TXTA28,S,R":PC=68:GOTO 9500

5200 D$="TXTA1,S,R":GOSUB 32650:GOTO 32500

5300 D$="TXTA2,S,R":GOSUB 32650:GOTO 32500

5400 D$="TXTA3,S,R":GOSUB 32650:GOTO 32500

5500 D$="TXTA4,S,R":GOSUB 32650:GOTO 32500

5600 D$="TXTA5,S,R":GOSUB 32650:GOTO 32500

5700 D$="TXTA6,S,R":GOSUB 32650:GOTO 32500

--------------------------------

Zischende Pfähle

5800 D$="TXTA7,S,R":GOSUB 32650:PC=47

5820 GOSUB 32580

5830 QW=QW+1:IF QW>=5 THEN 5300

5831 REM

5835 ZL=5840:GOTO 32300

5840 IF W1<>8 OR W2<>48 THEN GOSUB 32740:GOTO 5830

5855 REM

5860 GOTO 6000

schnee unter dir weg

6000 D$="TXTA29,S,R":GOSUB 32650:PC=97

6020 GOSUB 32580

6030 ZL=6040:GOTO 32300

6040 IF W1<>23 OR W2<>32 THEN GOSUB 32740:GOTO 6030

6045 GOSUB 32765

in der Stadt

6100 D$="TXTB19,S,R":GOSUB 32650:PC=99:GOSUB 32580

6130 ZL=6140:GOTO 32300

6135 IF W1=31 AND W2=48 THEN 6150

6140 IF W1=31 AND W2=48 THEN 6150

6145 GOSUB 32740:GOTO 6130

6150 GOSUB 32765:POKE 214,YD+6:SYS 58640

6155 PRINT " EIN  PLOETZLICH AUFTAUCHENDER  SCHNEE-"

6160 PRINT " STURM  HAT DICH AN  EINEN  ANDEREN ORT"

6165 PRINT " VERWEHT....":GOSUB 2

6180 GOTO 4200

Tante-Emma-Laden

6200 D$="TXTB20,S,R":GOSUB 32650:PC=100

6220 GOSUB 32580

6230 ZL=6240:GOTO 32300

6240 IF W1<>7 OR ASC(MID$(G$,49,1)+CHR$(0))<>0 THEN 40080

6245 IF W2=53 THEN GOTO 6260
wenn nirgendwo anders im Adv ein Messer, dann hier :))

6250 IF W2<55 OR W2>60 THEN GOSUB 32740:GOTO 6230

6260 G$=LEFT$(G$,48)+CHR$(200)+RIGHT$(G$,LEN(G$)-49)

6265 G$=LEFT$(G$,W2-1)+CHR$(0)+RIGHT$(G$,LEN(G$)-W2)

6270 GOTO 6230

Bar von innen

6300 D$="TXTB21,S,R":PC=101:GOTO 9500

Theke in der Bar

6400 D$="TXTB24,S,R":GOSUB 32650:PC=104

6420 GOSUB 32580

6430 ZL=6440:GOTO 32300

6440 IF W1<>21 OR W2<>68 OR ASC(MID$(G$,49,1)+CHR$(0))<>0 THEN 40090
BESTELLEN + GRAMWUMLS + Geld
6450 D$="TXTB28,S,R":GOSUB 32650
Gramwumls getrunken
6455 REM

6460 GOTO 32500

Hinterzimmer der Bar

6500 D$="TXTB25,S,R":GOSUB 32650:PC=105

6520 GOSUB 32580

6530 ZL=6540:GOSUB 32300

6540 IF W1<>11 OR W2<>67 THEN GOSUB 32740:GOTO 6530

6550 D$="TXTB26,S,R":GOSUB 32650:GOSUB 32765
Werbezettel betrachtet
6555 POKE 214,YD+6:SYS 58640:GOSUB 2

6560 GOSUB 32765:GOTO 6500

Toilette in der Bar

6600 D$="TXTB22,S,R":GOSUB 32650:PC=102

6620 GOSUB 32580

6630 ZL=6640:GOTO 32300

6640 IF W1<>22 OR W2<>69 THEN GOSUB 32740:GOTO 6630

6650 D$="TXTB23,S,R":GOSUB 32650:GOSUB 32765
Gekritzel gelesen
6655 POKE 214,YD+6:SYS 58640:GOSUB 2

6660 GOSUB 32765:GOTO 6600

in der Stadt (im NW allgemein zugängliches Gebäude

6700 D$="TXTB17,S,R":PC=97:GOTO 9500

in der Stadt am Nordpol (Anfangspunkt)

6800 D$="TXTB18,S,R":PC=98:GOTO 9500

verlaufen (Neubeginn)

6900 D$="TXTB9,S,R":GOSUB 32650:GOTO 32500

Schneesturm (Neubeginn)

7000 D$="TXTB10,S,R":GOSUB 32650:GOTO 32500

im Osten Polizeigebäude

7100 D$="TXTB13,S,R":PC=93:GOTO 9500

im Westen Friedhof

7200 D$="TXTB14,S,R":PC=94:GOTO 9500

--------------------------------

im Friedhof

(SG=0 – nur Friedhof, SG=1 – Sarg ausgegraben, SG=2 – Sarg geöffnet)

7300 D$="TXTB15,S,R":IF SG>0 THEN D$="TXTB29,S,R"
Bild „Friedhof“. wenn SG>0 dann Bild “geschlossener Sarg”
7305 IF SG=2 THEN D$="TXTB27,S,R"
wenn SG=2 dann Bild “geöffneter Sarg mit Scott”
7310 GOSUB 32650:PC=95
Text+Bild laden
7320 GOSUB 32580
mögliche Richtungen anzeigen
7330 ZL=7332:GOTO 32300
auf Eingabe warten
7332 IFW1=2ANDW2=43ANDASC(MID$(G$,53,1)+CHR$(0))<>0 AND SG=1 THEN GOSUB 40100


     ( wenn  OEFFNEN + SARG + kein Messer im Inv + Sarg ist ausgegraben und zu, dann „der Deckel klemmt“ )

7334 IFW1=2ANDW2=43ANDASC(MID$(G$,53,1)+CHR$(0))=0 AND SG=1 THEN 40120

     ( wenn  OEFFNEN + SARG + Messer im Inv + Sarg ist ausgegraben und zu, dann „Schlüssel gef.(im Inv)+Sarg offen )
7336 IFW2=68ANDW1=2ANDSG=0 AND ASC(MID$(G$,47,1)+CHR$(0))=0 THEN SG=1:GOTO 7300

     ( wenn OEFFNEN + GRAB + Schaufel im Inv,  dann “Sarg ausgegraben” )

7340 IF W1<>11 OR W2<>68 THEN GOSUB 32740:GOTO 7330

     ( wenn GRAB + SEHEN dann “Grabsteininschrift zeigen” )

7350 D$="TXTB16,S,R":GOSUB 32650:GOSUB 32765:POKE 214,YD+6:GOSUB 2

7360 GOSUB 32765:GOTO 7300

--------------------------------

gebilde im norden koennten haeuser sein. im osten was sehr wichtig

7900 D$="TXTB5,S,R":PC=85:GOTO 9500

--------------------------------

verlassene stadt am totalen nullpunkt

8000 D$="TXTB4,S,R":PC=84:GOTO 9500

in einer halbzervallenen Holzbaracke vor Holzbohle mit Diamant

8100 D$="TXTB1,S,R":GOSUB 32650:PC=81

8120 GOSUB 32580

8130 ZL=8140:GOTO 32300
auf Eingabe warten
8140 IF W1<>1 OR W2<>66 THEN GOSUB 32740:GOTO 8130

8145 REM

8150 D$="TXTB2,S,R":GOSUB 32650:GOTO 32500
Haus eingestürzt und ENDE

in der alten Holzbaracke

(wenn AND TH=1 nicht in Zeilen 8240 und 8241 eingefügt ist, kann man die Truhe ohne sie augegraben

zu haben öffnen und das Spiel beenden  -  wohl nicht im Sinne des Programmierers)  :))
8200 D$="TXTB3,S,R":IF TH=1 THEN D$="TXTB32,S,R"
Wenn TH=1 dann alte Holzbaracke mit geschlssener Truhe
8210 PC=83:GOSUB 32650:GOSUB 32580

8230 ZL=8240:GOTO 32300

8240 IF W1=2 AND W2=65 AND TH=1 AND ASC(MID$(G$,71,1)+CHR$(0))<>0 THEN GOSUB 40150

     (OEFFNEN + TRUHE + Schlüssel nicht im Inventar) – Sie ist verschlossen

8241 IF W1=2 AND W2=65 AND TH=1 ASC(MID$(G$,71,1)+CHR$(0))=0 THEN 8250

     (OEFFNEN + TRUHE + Schlüssel im Inventar) – Schlussbild – Spiel gelöst
8242 IFW2<>68ORW1<>0ORASC(MID$(G$,47,1)+CHR$(0))<>0 THEN GOSUB 32740:GOTO 8230

     (GRABEN + Schaufel im Inventar) – TH=1 (Truhe wird sichbar)

8245 REM

8246 TH=1:GOTO 8200

8250 REM

8255 REM

8260 GOSUB 32762:SYS 51715:OPEN 1,8,2,"ENDE,P,R":CLOSE 1

8265 OPEN 1,8,15:INPUT#1,XY:CLOSE1 :IF XY<>0 THEN 8260

8270 LOAD"ENDE",8

====================================================

Polizeigebäude

8300 REM

8310 D$="TXTB11,S,R":GOSUB 32650:PC=91

8320 GOSUB 32580

8325 IF ASC(MID$(G$,70,1)+CHR$(0))<>0 THEN 8350

8330 ZL=8340:GOTO 32300

8340 GOSUB 32740:GOTO 8330

8350 GOSUB 32765:POKE 214,YD+6:GOSUB 2

8360 D$="TXTB12,S,R":GOSUB 32650:GOTO 32500

---------------------------------------

Irrweg zur Hütte E1-7 “ gaehn, schnarch...“  -(N=8800 !!!!, O=9150,W=8450)

8400 D$="TXTB31,S,R":PC=152:GOTO 9500

Irrweg zur Hütte E1-6 “weisst du jetzt, woher das spiel seinennamen hat“  -(O=8400,W=8500)

8450 D$="TXTB8,S,R":PC=153:WW=8500:OO=8400:GOTO 9600

Irrweg zur Hütte E1-5 “schon wieder in einer endlosen schneewueste! (kicher...)“  -(O=8450,W=8550)

8500 D$="TXTB6,S,R":PC=154:WW=8550:OO=8450:GOTO 9600

Irrweg zur Hütte E1-4 “immer dieser Schnee“  -(O=8500,W=8600)

8550 D$="TXTB7,S,R":PC=155:WW=8600:OO=8500:GOTO 9600

Irrweg zur Hütte E1-3 “gahn schnarch“  -(O=8550, W=8650, N=9000)

8600 D$="TXTB31,S,R":GOSUB 32650:PC=156

8620 WE=8650:OT=8550:ND=9000:GOSUB 32580

8630 ZL=8640:GOTO 32300

8640 GOSUB 32740:GOTO 8630

Irrweg zur Hütte E1-2 „weist du jetzt woher das Spiel seinen Namen hat“  -(O=8600, W=9100)

8650 D$="TXTB8,S,R":PC=157:GOTO 9500

8670 GOSUB 32580

8680 ZL=8690:GOTO 32300

8690 GOSUB 32740:GOTO 8680

Irrweg zur Hütte E2-6 “gaehn, schnarch...“  -(N=7900, O=9050, W=8750)
8700 D$="TXTB31,S,R":GOSUB 32650:PC=158

8720 ND=7900:WE=8750:OT=9050:GOSUB 32580

8730 ZL=8740:GOTO 32300

8740 GOSUB 32740:GOTO 8730

Irrweg zur Hütte E2-5 “weisst du jetzt, woher das spiel seinennamen hat“  -(O=8700, W=8800)

8750 D$="TXTB6,S,R":PC=159:WW=8800:OO=8700:GOTO 9600

Irrweg zur Hütte E2-4 „den text kann man sich sparen“  -(S=8400, O=8750, W=8850)
8800 D$="TXTB30,S,R":PC=160:GOTO 9500

Irrweg zur Hütte E2-3 „immer dieser schnee“  -(O=8800, W=8900)
8850 D$="TXTB7,S,R":PC=170:WW=8900:OO=8800:GOTO 9600

8900 D$="TXTB8,S,R":PC=171:WW=8950:OO=8850:GOTO 9600

Irrweg zur Hütte E2-2 “schon wieder in einer endlosen schneewueste! (kicher...)  -(O=8850, W=8950)

8950 D$="TXTB6,S,R":PC=172:WW=9000:OO=8900:GOTO 9600

Irrweg zur Hütte E2-1 “ den text kann man sich sparen“  -(O=8950, W=9050, S=8600)

9000 D$="TXTB30,S,R":GOSUB 32650:PC=173

9020 WE=9050:OT=8950:SU=8600:GOSUB 32580

9030 ZL=9040:GOTO 32300

9040 GOSUB 32740:GOTO 9030

Irrweg zur Hütte E2-7 “immer dieser schnee“  -( O=9000, W=8700)
9050 D$="TXTB7,S,R":PC=174:WW=8700:OO=9000:GOTO 9600

--------------------------------

Irrweg zur Hütte E1-1 „schon wieder in einer endlosen schneewueste! (kicher...)“  -(O=8650, W=9150

9100 D$="TXTB6,S,R":GOSUB 32650:PC=150
Bild laden und anzeigen


9120 SU=6800: GOSUB 325
0
man beachte die erste „Null“ bei 325
0. Das ergibt einen

Undefn’t Statement Error, da für Interpeter Zeilennr. 3250


(Die gibt’s im Programm nicht), und einen schönen bleuen


Bildschirm mit einem hübschen READY  :))


Nun steht aber davor SU=6800 und das heißt Richtung


Süden geht’s mit der Zeilennummer 6800 weiter.


Es sollte 32580 lauten da dieser Einsprung die möglichen


Richtungen anzeigt (Ich hab’s auch so geändert)


(Variablen im Text = O, W) + nun auch S


DIESER FEHLER IST FATAL UND MACHT DAS SPIEL UNLÖSBAR!! 


Denn genau hier muß man weiter, damit man das Spiel beenden kann !!


DAS HAUPTPRAMM TRÄGT DEN NAMEN „DEMOFILE“ ZU RECHT !!
9130 ZL=9140:GOTO 32300
auf Eingabe warten

9140 GOSUB 32740:GOTO 9130
Eingabemaske löschen und 1 Zeile zurück (auf Eingabe warten)

--------------------------------

Irrweg zur Hütte E1-8 „immer dieser schnee.“  -(O=9100,W=8400)

9150 D$="TXTB7,S,R":GOSUB 32650:PC=151

9170 GOSUB 32580
mögliche Richtungen anzeigen
9180 ZL=9190:GOTO 32300
auf Eingabe warten
9190 GOSUB 32740:GOTO 9180
verlegenheitsantwort und 1 Zeile zurück
--------------------------------
Unterprogramm zu “Irrweg zu Hütte“ Ebene 2
9500 GOSUB 32650


9520 GOSUB 32580

9530 ZL=9540:GOTO 32300

9540 GOSUB 32740:GOTO 9530

--------------------------------
Unterprogramm zu “Irrweg zu Hütte“ Ebene 1

9600 GOSUB 32650:WE=WW:OT=OO

9610 GOSUB 32580

9630 ZL=9640:GOTO 32300

9640 GOSUB 32740:GOTO 9630

--------------------------------
10000 GOSUB 32765:POKE 214,YD+6:SYS 58640

10010 PRINT " DEINE LEBENSZEIT IST ABGELAUFEN!!!"

10020 GOSUB 2

10030 GOTO 32500

29000 REM

29060 D2$="GAME,S,W"

29065 CLOSE 1:CLOSE 2:GOSUB 32765:POKE 214,YD+5:SYS 58640

29066 PRINT:GOSUB 40200

29069 SYS 51715:B=1:IF W1=20 THEN B=2

29070 ON B GOTO 29100,29200

29090 CLOSE 1:SYS 51715:POKE 53281,9:SYS 51712

29092 GOSUB 32765:POKE 214,YD+5:SYS 58640:PRINT:GOSUB 40210

29095 OPEN 2,8,2,"D:TXT*,S,R":CLOSE2:OPEN 1,8,15:INPUT#1,A:CLOSE1

29096 IF A<>0 THEN 29090

29097 IF W1=20 THEN GOTO ZL

29098 IF W1=19 THEN GOTO ZL-40

29100 D2$="GAME,S,R"

29110 OPEN 1,8,2,D2$:GOSUB 29125:LO=A:GOSUB 29125:HI=A

29115 ZL=HI*256+LO

29116 INPUT#1,G$

29117 FOR I=40 TO 71:A=ASC(MID$(G$,I,1)+CHR$(0))

29118 IF A=199 THEN G$=LEFT$(G$,I-1)+CHR$(0)+RIGHT$(G$,LEN(G$)-I)

29119 NEXT

29120 GOSUB 29125:LK=A:GOSUB 29125:GG=A:GOSUB 29125:ZU=A

29121 GOSUB 29125:QW=A:GOSUB 29125:SG=A:GOSUB 29125:TH=A

29122 GOSUB 29125:PC=A:GOTO 29127

29125 GET#1,A$:GET#1,B$:IF A$="" THEN A$=CHR$(0)

29126 A=ASC(A$):RETURN

29127 INPUT#1,D$

29128 A=LEN(D$):D$="TXT"+RIGHT$(D$,LEN(D$)-3)

29130 CLOSE 1:FL=1:OPEN 1,8,15:INPUT#1,A:CLOSE 1:IF A<>0 THEN 29065

29180 GOTO 29090

29200 GOSUB 32765:POKE 214,YD+6:SYS 58640

29201 PRINT " DATEN-DISK FORMATIEREN ? (J/N)"

29202 WAIT 203,63:GETA$

29203 IF A$<>"J" THEN 29210

29205 OPEN 2,8,15,"N0:DATENDISK,00":CLOSE2:OPEN 2,8,15:INPUT#2,A:CLOSE2

29206 IF A<>0 THEN 29065

29210 D2$="@:"+D2$

29213 OPEN 1,8,2,D2$:HI=INT(ZL/256):LO=ZL-HI*256

29215 PRINT#1,CHR$(LO):PRINT#1,CHR$(HI)

29216 FOR I=40 TO 71:A=ASC(MID$(G$,I,1)+CHR$(0))

29217 IF A=0 THEN G$=LEFT$(G$,I-1)+CHR$(199)+RIGHT$(G$,LEN(G$)-I)

29218 NEXT

29219 PRINT#1,G$

29220 FOR I=40 TO 71:A=ASC(MID$(G$,I,1)+CHR$(0))

29221 IF A=199 THEN G$=LEFT$(G$,I-1)+CHR$(0)+RIGHT$(G$,LEN(G$)-I)

29222 NEXT:PRINT#1,CHR$(LK):PRINT#1,CHR$(GG):PRINT#1,CHR$(ZU)

29225 PRINT#1,CHR$(QW):PRINT#1,CHR$(SG):PRINT#1,CHR$(TH)

29226 PRINT#1,CHR$(PC):PRINT#1,D$

29230 CLOSE 1:FL=1:OPEN 2,8,15:INPUT#2,A:CLOSE2:IF A<>0 THEN 29065

29250 GOTO 29090

31000 RESTORE 10

31020 FOR I=1 TO GS:READ A,WR$

31030 IF W2=A AND ASC(MID$(G$,A,1)+CHR$(0))=0 THEN GOSUB 40300

31040 NEXT I

31050 RETURN

31100 RESTORE 10:GOSUB 32765:POKE 214,YD+5:SYS 58640:PRINT

31110 PRINT "ZU SEHEN: ":Y=YD+7

31120 FOR I=1 TO GS:READ A,WR$

31130 IF Y>YD+10 THEN GOSUB 31180

31140 IFASC(MID$(G$,A,1)+CHR$(0))=PCTHENPOKE 214,Y:SYS58640:PRINTWR$:Y=Y+1

31150 NEXT:GOSUB 31180

31160 FL=1:RETURN

31180 POKE 214,YD+11:SYS 58640:GOSUB 2

31185 POKE 214,YD+6:SYS 58640:Y=YD+6:RETURN

32000 RESTORE 10

32011 IF GG=4 THEN GOSUB 40500:RETURN

32020 FOR I=1 TO GS:READ A,WR$

32030 IF W2=A AND ASC(MID$(G$,A,1)+CHR$(0))=PC AND GG<4 THEN 40510

32040 NEXT I

32050 RETURN

===================================

Inventar ansehen

32200 GOSUB32765:POKE214,YD+6:SYS 58640:PRINT"DU HAST:  ":Y=YD+7
Eingabemaske löschen Screen+Farbz. setzen
32203 RESTORE 10:FOR I=1 TO GS

32204 READ A,WR$

32205 IFASC(MID$(G$,A,1)+CHR$(0))=0THENPOKE214,Y:SYS58640:PRINTWR$:Y=Y+1

32206 NEXT :FL=1

32218 POKE 214,YD+10:SYS 58640:PRINT:GOSUB 2:RETURN

===================================

Warten auf Eingabe

------------------

32300 GOSUB 32765:A=INT(RND(0)*5):POKE214,YD+5:SYS 58640:PRINT:GOTO 32306+A

32306 PRINT " ALSO?":GOTO 32315

32307 PRINT " TJA ?":GOTO 32315

32308 PRINT " NUN ?":GOTO 32315

32309 PRINT " HMM ?":GOTO 32315

32310 PRINT " NAJA?":GOTO 32315

32315 POKE 53281,0:X=6:SP=1:FOR I=828 TO 828+39:POKE I,32:NEXT

32316 SC=0:FG=0:POKE 214,YD+6:SYS58640:PRINT TAB(X);"*"

32320 IF FG=1 THEN FG=0:IF ZC=32 THEN SC=SC-1

32322 WAIT 198,255:GETA$:A=ASC(A$+CHR$(0)):IF A=13 THEN 32350

32325 IF A=20 AND SP>1 THEN 40600

32330 IF (A<65 OR A>90) AND (A<>32 OR (A=32 AND SC=2)) THEN 32320

32340 POKE 214,YD+6:SYS 58640:PRINTTAB(X);"
";A$;"*‑":POKE 827+SP,A

32341 IF X<35 THEN X=X+1:SP=SP+1

32342 IF A=32 THEN SC=SC+1

32345 GOTO 32320

32350 AN=2:IF SC=2 THEN AN=3

32357 FL=0:GOSUB 40700

32358 IF W2>26 AND W2<33 AND W1=0 THEN W1=W2:W2=0

32359 ZU=ZU+1:IF ZU=150 THEN 10000

32360 IF W1=27 AND ND THEN GOTO ND

32361 IF W1=28 AND SU THEN GOTO SU

32362 IF W1=29 AND OT THEN GOTO OT

32363 IF W1=30 AND WE THEN GOTO WE

32364 IF W1=31 AND HC THEN GOTO HC

32365 IF W1=32 AND DN THEN GOTO DN

32366 IF W1=1 THEN GOSUB 32000

32367 IF W2=33 THEN GOSUB 32200

32368 IF W1=10 THEN GOSUB 31000

32369 IF W2=11 AND W1=0 THEN GOSUB 31100

32370 IF (W1=19 OR W1=20) AND W2=62 THEN 29000

32399 GOTO ZL

32400 DK=1:PRINT"“":SYS 51715:POKE 51483,71

32405 POKE 51722,144

32410 SYS 51718,"PIC1",8

32455 GOSUB 32765:POKE 214,YD:SYS 58640:PRINT

32456 POKE 788,52:POKE792,193:POKE793,254

---  Kopierschutz ---

32457 OPEN 1,8,15:OPEN 2,8,2,"#":PRINT#1,"U1 2 0 1 10":INPUT#1,A

32459 A=23 : IF A<>23 THEN SYS 42564
wenn A nicht 23 dann Einsprung Basic NEW-Routine
32460 CLOSE 1:CLOSE 2

---------------------

32461 PRINT "
 HERZLICH WILLKOMMEN BEI  ZIELPUNKT: 0' ";

32465 PRINT " NORD. WIR HOFFEN, DASS IHNEN DAS SPIEL ";

32470 PRINT " SPASS MACHEN WIRD."

32475 PRINT

32476 PRINT "        KEMAL EZCAN & AXEL WEBER"

32480 POKE 53265,PEEK(53265)OR16:POKE 53281,9:SYS 51712

32500 POKE 214,YD:SYS 58640

32510 PRINTTAB(0);"ž
 DRUECKE SPACE UM DAS SPIEL ZU BEGINNEN "

32515 GOSUB 41000
Eigabemaske löschen
32520 GETA$:IFA$<>" " THEN 32520

32522 GG=0:ZU=0:QW=0:SG=0:TH=0:LK=0

32525 POKE 214,YD:SYS 58640:PRINT "

  MOEGLICHE RICHTUNGEN:                 ’"

Inventar vorbelegen, wenn Gegenstand genommen wird an der Stelle, die der Nummer des Gegenstands

Entspricht eine „0“ eingesetzt

32530 G$="ÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈÈ"

32532 RESTORE 32540

32535 FOR I=40 TO 71:READ A:G$=G$+CHR$(A):NEXT

32540 DATA 2,200,10,200,200,200,200,62,200,63,1,3,52,81,67,200

32542 DATA 200,200,200,200,200

32545 DATA 200,200,200,200,200,200,200,200,200,8,10

32560 GOTO 1000

=========================================

mögliche Richtungen anzeigen

32580 POKE 214,YD:SYS 58640:PRINTTAB(26);"

          "

32583 POKE 214,YD:SYS 58640:PRINTTAB(26);

32584 IF ND<>0 THEN PRINT "
N ’";

32586 IF SU<>0 THEN PRINT "
S ’";

32588 IF OT<>0 THEN PRINT "
O ’";

32590 IF WE<>0 THEN PRINT "
W ’";

32592 IF HC<>0 THEN PRINT "
H ’";

32594 IF DN<>0 THEN PRINT "
R ’";

32596 RETURN

=========================================

32600 DIM WI$(40),D$(20),WR$(35),G$(71),D2$(20)

32601 RESTORE 32602:FOR I=0 TO 19:READ A:POKE 1024+I,A:NEXT I

32602 DATA 72,138,72,169,10,162,18,141

32610 DATA 10,212,141,23,208,142,24,208,104,170,104,64

32630 GOSUB 42000

32640 RETURN

=========================================

32650 REM TEXT & BILD LADEN
SYS 51718 = Laderoutine, SYS 51715 = Rasterinterupt aus, SYS 51712 = Rasterinterrupt ein

---------------------------

32652 SYS 51715:IF RIGHT$(D$,4)<>",S,R" THEN D$=D$+",S,R"

32653 D$="PIC"+MID$(D$,4,LEN(D$)-7):SYS 51718,D$,8:OPEN 15,8,15:GET#15,XY

32654 D$="TXT"+RIGHT$(D$,LEN(D$)-3)+",S,R":CLOSE 15:IF XY<>0 THEN 32666

32655 OPEN 2,8,2,D$:FOR I=0 TO 199:GET#2,A$:POKE 1584+I,ASC(A$):NEXT:GET#2,A$

32656 INPUT#2,A1:INPUT#2,A2:ND=A2*256+A1
Highbyte+Lowbyte Zeilennr für N einlesen (Znr=LBx256+HB)
32657 INPUT#2,A1:INPUT#2,A2:SU=A2*256+A1
Highbyte+Lowbyte Zeilennr für S einlesen (Znr=LBx256+HB)
32658 INPUT#2,A1:INPUT#2,A2:OT=A2*256+A1
Highbyte+Lowbyte Zeilennr für O einlesen (Znr=LBx256+HB)
32659 INPUT#2,A1:INPUT#2,A2:WE=A2*256+A1
Highbyte+Lowbyte Zeilennr für W einlesen (Znr=LBx256+HB)
32660 INPUT#2,A1:INPUT#2,A2:HC=A2*256+A1
Highbyte+Lowbyte Zeilennr für Hoch einlesen (Znr=LBx256+HB)
32661 INPUT#2,A1:INPUT#2,A2:DN=A2*256+A1
Highbyte+Lowbyte Zeilennr für Runter einlesen (Znr=LBx256+HB)
32665 CLOSE 2:FOR I=0 TO 239:POKE 55296+520+I,1:NEXT I

32666 POKE 53281,9

32668 OPEN1,8,15:INPUT#1,XY:CLOSE1

32669 IF XY=0 THEN SYS 51712:RETURN

32670 REM 
wenn Bild nicht gefunden dann
32675 SYS 51712:SYS 51715:CLOSE 2:GOSUB 32762:GOTO 32650
Diskette wenden und nochmal
=========================================

Bildschirm löschen

32700 REM

32710 FOR I=0 TO 24:PRINT"
                                        ";:NEXT

32730 RETURN

32740 IF FL=0 THEN GOSUB 32765:POKE 214,YD+10:PRINT:SYS 58640:GOSUB 40800:RETURN

32741 RETURN

=========================================
Verlegenheitsantworten

32742 PRINT " DAS GEHT NICHT!";:RETURN

32743 PRINT " UEBERLEG' NOCH MAL!";:RETURN

32744 PRINT " WAS SOLL DENN DAS?";:RETURN

32745 PRINT " SONST NOCH WAS?";:RETURN

32746 PRINT " NOCH'N WUNSCH?";:RETURN

32747 PRINT " DIR GEHT'S WOHL ZU GUT!!";:RETURN

=========================================

Diskette wenden wenn File nicht gefunden

32762 GOSUB 32765:POKE 214,YD+6:SYS 58640:PRINT " BITTE DIE DISK. WENDEN"

32763 GOSUB 2:GOSUB 32765:RETURN
auf Space-Taste warten
32765 GOSUB 41000:RETURN
Eingabemaske löschen
=========================================

Unterprogramm zu “Karte am Dachboden“

GG=Gegenstadsanzahl, G$=Inventar

40000 GOSUB 32765:POKE 214,YD+5:SYS 58640:PRINT
Eingabefeld löschen
40001 PRINT"DIE KARTE IST ZU STAUB ZERFALLEN."
1 Gegenstand weniger, Karte zerfallen
40005 GG=GG-1:G$=LEFT$(G$,41)+CHR$(200)+RIGHT$(G$,LEN(G$)-42)
in G$ an 42. Stelle Chr$200 einfügen
40010 GOSUB 2:RETURN
auf SPACE warten und zurück
=========================================

40020 GOSUB 1680:FL=1:GOTO 1600

40030 GOSUB 32740:GOTO 2130

40040 POKE 214,YD+6:SYS58640:PRINT " DU HAST SCHON VIER GEGENSTAENDE!":GOTO 2175

40050 GOSUB 32740:GOTO 3030

40060 GOSUB 32740:GOTO 3130

40070 GOSUB 32740:GOTO 3330

40080 GOSUB 32740:GOTO 6230

40090 GOSUB 32740:GOTO 6430

---------------------------

Aufruf von Bild „Friedhof“ Zeile 7332 wenn - OEFFNEN+SARG+kein Messer im Inv+SG=1
40100 GOSUB 32765:POKE 214,YD+6:SYS 58640

40110 PRINT " DER DECKEL KLEMMT!":GOSUB 2:FL=1:RETURN

---------------------------

Aufruf von Bild „Friedhof“ Zeile 7334 wenn - OEFFNEN+SARG+Messer im Inv+SG=1
40120 SG=2:G$=LEFT$(G$,70)+CHR$(95)+RIGHT$(G$,LEN(G$)-71):GOTO 7300  FALSCH!! (Nicht 95 sondern 0)
40120 SG=2:G$=LEFT$(G$,70)+CHR$(0)+RIGHT$(G$,LEN(G$)-71):GOTO 7300
Schlüssel im Inventar

---------------------------

Aufruf von Bild “in der alten Holbaracke“ Zeile 8240 wenn - OEFFNEN + TRUHE + Schlüssel nicht im Inventar
40150 GOSUB 32765:POKE 214,YD+6:SYS 58640:PRINT " SIE IST VERSCHLOSSEN!"

40160 GOSUB 2:FL=1:RETURN

---------------------------

40200 PRINT " BITTE IHRE DATEN-DISK EINLEGEN":GOSUB 2:GOSUB 32765:RETURN

40210 PRINT " BITTE DIE PROGRAMMDISK. EINLEGEN":GOSUB 2:GOSUB 32765:RETURN

40300 GG=GG-1:G$=LEFT$(G$,A-1)+CHR$(PC)+RIGHT$(G$,LEN(G$)-A):PRINT

40310 PRINT" O.K.":FL=1:FOR W=1 TO 100:NEXT:RETURN

40500 PRINT :PRINT " DU TRAEGST ZUVIEL!!":FOR W=1 TO 20:NEXT W:FL=1:RETURN

40510 GG=GG+1:G$=LEFT$(G$,A-1)+CHR$(0)+RIGHT$(G$,LEN(G$)-A)

40520 PRINT:PRINT" O.K.":FL=1:FOR W=1 TO 500:NEXT:RETURN

40600 SP=SP-1:POKE 214,YD+6:SYS 58640:X=X-1:A=PEEK(827+SP):POKE 827+SP,32

40605 IF A=32 THEN SC=SC-1

40610 PRINTTAB(X);"
* ":FG=1:GOTO 32320

40700 POKE 50945,72:POKE 50950,2:IF SC=2 THEN POKE 50950,3

40701 POKE 828+SP,32:POKE 829+SP,32

40702 SYS 50944

40705 W2=PEEK(1018):W1=PEEK(1019)

40706 IF W1=255 THEN W1=0

40707 IF W2=255 THEN W1=0

40708 IF W2>26 AND W2<33 AND W1=0 THEN W1=W2:W2=0

40710 RETURN

40800 A=INT(RND(0)*6)+1:GOSUB 32741+A:FOR W=1 TO 150:NEXT :RETURN

===========================================

Eingabemaske löschen

SYS 58649 = Zeiger auf Bildschirm und Farbram aktualisieren

-------------------------------------------

41000 POKE 214,YD+5:SYS 58640:PRINT:FOR I=1 TO 4

41005 PRINT"
                                        ";:NEXT

41010 FOR I=1944 TO 1944+79:POKE I,32:NEXT:GOSUB 60000

41020 POKE 214,YD+6:SYS 58640:RETURN

===========================================

42000 REM GOTO,GOSUB,RESTORE

42005 SYS 900

42010 A=10*4096+8*256+10*16

42020 POKE A,32:POKE A+1,192:POKE A+2,2

42030 A=704

42040 POKE A,32:POKE A+1,138:POKE A+2,173:POKE A+3,76:POKE A+4,247:POKE A+5,183

42050 A=2*256+12*16+6

42060 POKE A,208:POKE A+1,3

42070 POKE A+2,76:POKE A+3,29

42080 POKE A+4,168:POKE A+5,32

42090 POKE A+6,192:POKE A+7,2

42100 POKE A+8,32:POKE A+9,19

42110 POKE A+10,166:POKE A+11,56

42120 POKE A+12,165:POKE A+13,95

42130 POKE A+14,233:POKE A+15,1

42140 POKE A+16,164:POKE A+17,96

42150 POKE A+18,76:POKE A+19,36

42160 POKE A+20,168:POKE 40996,197:POKE 40997,2:POKE 1,54

42170 RETURN

==========================================

Externe Programme + Verben laden

45000 LOAD"RAST*",8,1

45002 LOAD"WKENN*",8,1

45003 LOAD"VERB*",8,1

45004 LOAD"BASICMOVE",8,1

60000 RETURN

READY.

